

FUNCIONES Y SUBPROCESOS

Implementación de funciones y subprocesos en PSeInt

```
1  // FUNCIONES Y SUBPROCESOS
2  // Las funciones y subprocesos, permiten reutilizar partes de un algoritmo varias veces dentro del mismo
3  // algoritmo. Reduciendo el número de líneas de los algoritmos, estructurándolos mejor y optimizándolos
4  // Esto se hace por medio del paso de parámetros o argumentos desde el proceso principal hacia las funciones
5  // o subprocesos. Es recomendable y muy importante tener en cuenta que los nombres dados internamente a las
6  // variables que entran como parámetros o argumentos, deben ser distintos a los nombres de las variables
7  // usadas en el proceso principal.
8  // Las funciones devuelven un sólo valor en una variable denominada "variable de retorno".
9  // Los subprocesos en cambio permiten pasar o devolver al proceso principal, más de una variable que haya
10 // sido transformada dentro del subproceso, siempre y cuando se haya definido del tipo "Por Referencia"
11 // A continuación mostramos ejemplos sencillos de funciones y subprocesos
12
13 // Definición de la función suma, la cual suma los valores de argumento1 más argumento2,
14 // y el resultado lo asigna a la variable de retorno s
15 //
16 SubProceso s <- suma ( argumento1,argumento2 )
17     s<-argumento1+argumento2
18
19 Fin SubProceso
20
21
22 // Definición del subproceso sumar, el cual asigna a la variable argumento3 la suma los valores
23 // de argumento1 más argumento2. Observe que argumento3 está definido Por Referencia, permitiendo
24 // que el resultado de la operación pueda ser conocido por el proceso principal.
25 //
26 SubProceso sumar ( argumento1,argumento2, argumento3 Por Referencia )
27     argumento3<-argumento1+argumento2
28
29 Fin SubProceso
30
31 // Definición de la función producto, la cual multiplica los valores de argumento1 por argumento2,
32 // y el resultado lo asigna a la variable de retorno p
33 //
34 SubProceso p <- producto ( argumento1,argumento2 )
35     p<-argumento1*argumento2
36
37 Fin SubProceso
38
39 // Definición del subproceso multiplicar, el cual asigna a la variable argumento3 la multiplicación
40 // de los valores del argumento1 por el argumento2. Observe que argumento3 está definido Por Referencia,
41 // permitiendo que el resultado de la operación pueda ser conocido por el proceso principal.
42 //
43 SubProceso multiplicar ( argumento1,argumento2, argumento3 Por Referencia )
44     argumento3<-argumento1*argumento2
45
46 Fin SubProceso
47
48 Proceso Funciones_Y_Subprocesos
49     // Se solicitan los valores de a y b que serán pasados como parámetros a las funciones y subprocesos
50     Escribir sin saltar "Digite el valor el valor de a"
51     leer a
52     Escribir sin saltar " Digite el valor de b"
53     leer b
54     // Se inicializa la variable c en cero, que será utilizada para alacénar el resultado de ejecutar los
55     // subprocesos y será pasada a los subprocesos como el tercer parámetro o argumento
56     c<-0
57
58     // La manera de ejecutar una función es asignándola a una variable. En este caso se hace el llamado a
59     // la función suma, asignándola a la variable sumatoria. Los parámetros o argumentos que entran a la
60     // función son las variables a y b. Durante la ejecución de la función suma, estas variables tomarán el
61     // lugar de argumento1 y argumento2 respectivamente.
62     sumatoria<-suma(a,b)
63     Escribir "el resultado de la función suma es:",sumatoria
64
65     // En este caso se hace el llamado a la función producto, asignándola a la variable multiplicacion.
66     // Los parámetros o argumentos que entran a la función son las variables a y b. Durante la ejecución
67     // de la función producto, estas variables tomarán el lugar de argumento1 y argumento2 respectivamente.
68     multiplicacion<-producto(a,b)
69     Escribir "el resultado de la función producto es: ",multiplicacion
```

FUNCIONES Y SUBPROCESOS

```
70
71 // La manera de ejecutar un subproceso es colocando el nombre del subproceso con el conjunto de parámetros
72 // o argumentos necesarios para su ejecución. En este caso el subproceso sumar tiene los parámetros a,b
73 // y c. Durante la ejecución del subproceso sumar, estas variables tomarán el lugar de argumento1,
74 // argumento2 y argumento3 respectivamente. Para este caso en particular la variable c almacenará el
75 // resultado de ejecutar el subproceso sumar.
76 sumar(a,b,c)
77 Escribir "el resultado del subproceso sumar es: ",c
78
79 // En este caso el subproceso multiplicar tiene los parámetros a,b y c. Durante la ejecución del
80 // subproceso multiplicar, estas variables tomarán el lugar de argumento1, argumento2 y argumento3
81 // respectivamente. Para este caso en particular la variable c almacenará el resultado de ejecutar el
82 // subproceso multiplicar.
83 multiplicar(a,b,c)
84 Escribir "el resultado del subproceso multiplicar es:",c
85
86 FinProceso
```